/*
 * Copyright (c) 2013-2022 Apple Inc. All rights reserved.
 *
 * @APPLE_OSREFERENCE_LICENSE_HEADER_START@
 *
 * This file contains Original Code and/or Modifications of Original Code
 * as defined in and that are subject to the Apple Public Source License
 * Version 2.0 (the 'License'). You may not use this file except in
 * compliance with the License. The rights granted to you under the License
 * may not be used to create, or enable the creation or redistribution of,
 * unlawful or unlicensed copies of an Apple operating system, or to
 * circumvent, violate, or enable the circumvention or violation of, any
 * terms of an Apple operating system software license agreement.
 *
 * Please obtain a copy of the License at
 * http://www.opensource.apple.com/apsl/ and read it before using this file.
 *
 * The Original Code and all software distributed under the License are
 * distributed on an 'AS IS' basis, WITHOUT WARRANTY OF ANY KIND, EITHER
 * EXPRESS OR IMPLIED, AND APPLE HEREBY DISCLAIMS ALL SUCH WARRANTIES,
 * INCLUDING WITHOUT LIMITATION, ANY WARRANTIES OF MERCHANTABILITY,
 * FITNESS FOR A PARTICULAR PURPOSE, QUIET ENJOYMENT OR NON-INFRINGEMENT.
 * Please see the License for the specific language governing rights and
 * limitations under the License.
 *
 * @APPLE_OSREFERENCE_LICENSE_HEADER_END@
 */

/*
 *  Interface definition for the namespace facility.
 */

subsystem
#if KERNEL_USER
KernelUser
#endif /* KERNEL_USER */
vfs_nspace 867800;     /* 'V''N'00 */

#include <mach/std_types.defs>
#include <mach/mach_types.defs>

ServerPrefix receive_;
UserPrefix   send_;

/* DEPRECATED */
routine nspace_handle(
        nspace_handler_port : mach_port_t;
        pid                 : uint32_t;
     in path                : vfs_path_t;
    out handler_error       : int
);

routine nspace_resolve_cancel(
        nspace_handler_port : mach_port_t;
        req_id              : uint32_t
);

/*
 * nspace_resolve_path should not be moved even after it's not used, it's going
 * to mess up the calls numbering in mig.
 */
routine nspace_resolve_path(
        nspace_handler_port : mach_port_t;
        req_id              : uint32_t;
        pid                 : uint32_t;
        op                  : uint32_t;
     in path                : nspace_path_t;
    out xxx_rdar44371223    : int
);

/*
 * vfs_resolve_file is deprecated in favor of vfs_resolve_file_with_audit_token.
 */
routine vfs_resolve_file(
        nspace_handler_port : mach_port_t;
        req_id              : uint32_t;
        pid                 : uint32_t;
        op                  : uint32_t;
        offset              : int64_t;
        size                : int64_t;
     in path                : nspace_path_t
);

/*
 * vfs_resolve_dir is deprecated in favor of vfs_resolve_dir_with_audit_token.
 */
routine vfs_resolve_dir(
        nspace_handler_port : mach_port_t;
        req_id              : uint32_t;
        pid                 : uint32_t;
        op                  : uint32_t;
     in file_name           : nspace_name_t;
     in path                : nspace_path_t
);

routine vfs_resolve_file_with_audit_token(
        nspace_handler_port : mach_port_t;
        req_id              : uint32_t;
        op                  : uint32_t;
        offset              : int64_t;
        size                : int64_t;
     in path                : nspace_path_t;
     in req_atoken          : audit_token_t;	/* audit token of requester */
ServerAuditToken atoken     : audit_token_t
);

routine vfs_resolve_dir_with_audit_token(
        nspace_handler_port : mach_port_t;
        req_id              : uint32_t;
        op                  : uint32_t;
     in file_name           : nspace_name_t;
     in path                : nspace_path_t;
     in req_atoken          : audit_token_t;	/* audit token of requester */
ServerAuditToken atoken     : audit_token_t
);

routine vfs_resolve_reparent_with_audit_token(
        nspace_handler_port : mach_port_t;
        req_id              : uint32_t;
        op                  : uint32_t;
     in path                : nspace_path_t;
     in dest_path           : nspace_path_t;
     in req_atoken          : audit_token_t;	/* audit token of requester */
ServerAuditToken atoken     : audit_token_t
);

/* vim: set ft=c : */
